Symmetries of Heat Equation

$u_t = u_{xx}$

```
F = UT - UXX
ux = D[u[x, t], x];
ut = D[u[x, t], t];
utx = D[ut, x];
uxx = D[ux, x];
```

UT − UXX

```
SymmetryCondition = η01 D[F, UT] + η20 D[F, UXX]
```

$\eta 01 - \eta 20$

```
η01 = D[η[x, t, u[x, t]], t] - ux D[ξ1[x, t, u[x, t]], t] - ut D[ξ2[x, t, u[x, t]], t]
η10 = D[η[x, t, u[x, t]], x] - ux D[ξ1[x, t, u[x, t]], x] - ut D[ξ2[x, t, u[x, t]], x]
η20 = D[η10, x] - uxx D[ξ1[x, t, u[x, t]], x] - utx D[ξ2[x, t, u[x, t]], x]
```

$$u^{(0,1)}[x,t]\,\eta^{(0,0,1)}[x,t,u[x,t]] + \eta^{(0,1,0)}[x,t,u[x,t]] - u^{(1,0)}[x,t]\left(u^{(0,1)}[x,t]\,\xi 1^{(0,0,1)}[x,t,u[x,t]] + \xi 1^{(0,1,0)}[x,t,u[x,t]]\right) - u^{(0,1)}[x,t]\left(u^{(0,1)}[x,t]\,\xi 2^{(0,0,1)}[x,t,u[x,t]] + \xi 2^{(0,1,0)}[x,t,u[x,t]]\right)$$

$$u^{(1,0)}[x,t]\,\eta^{(0,0,1)}[x,t,u[x,t]] + \eta^{(1,0,0)}[x,t,u[x,t]] - u^{(1,0)}[x,t]\left(u^{(1,0)}[x,t]\,\xi 1^{(0,0,1)}[x,t,u[x,t]] + \xi 1^{(1,0,0)}[x,t,u[x,t]]\right) - u^{(0,1)}[x,t]\left(u^{(1,0)}[x,t]\,\xi 2^{(0,0,1)}[x,t,u[x,t]] + \xi 2^{(1,0,0)}[x,t,u[x,t]]\right)$$

$$u^{(2,0)}[x,t]\,\eta^{(0,0,1)}[x,t,u[x,t]] - 2u^{(2,0)}[x,t]\left(u^{(1,0)}[x,t]\,\xi 1^{(0,0,1)}[x,t,u[x,t]] + \xi 1^{(1,0,0)}[x,t,u[x,t]]\right) - 2u^{(1,1)}[x,t]\left(u^{(1,0)}[x,t]\,\xi 2^{(0,0,1)}[x,t,u[x,t]] + \xi 2^{(1,0,0)}[x,t,u[x,t]]\right) + u^{(1,0)}[x,t]\,\eta^{(1,0,1)}[x,t,u[x,t]] + u^{(1,0)}[x,t]\left(u^{(1,0)}[x,t]\,\eta^{(0,0,2)}[x,t,u[x,t]] + \eta^{(1,0,1)}[x,t,u[x,t]]\right) + \eta^{(2,0,0)}[x,t,u[x,t]] - u^{(1,0)}[x,t]\left(u^{(2,0)}[x,t]\,\xi 1^{(0,0,1)}[x,t,u[x,t]] + u^{(1,0)}[x,t]\,\xi 1^{(1,0,1)}[x,t,u[x,t]] + u^{(1,0)}[x,t]\left(u^{(1,0)}[x,t]\,\xi 1^{(0,0,2)}[x,t,u[x,t]] + \xi 1^{(1,0,1)}[x,t,u[x,t]]\right) + \xi 1^{(2,0,0)}[x,t,u[x,t]]\right) - u^{(0,1)}[x,t]\left(u^{(2,0)}[x,t]\,\xi 2^{(0,0,1)}[x,t,u[x,t]] + u^{(1,0)}[x,t]\,\xi 2^{(1,0,1)}[x,t,u[x,t]] + u^{(1,0)}[x,t]\left(u^{(1,0)}[x,t]\,\xi 2^{(0,0,2)}[x,t,u[x,t]] + \xi 2^{(1,0,1)}[x,t,u[x,t]]\right) + \xi 2^{(2,0,0)}[x,t,u[x,t]]\right)$$

```
SymmetryCondition =
  SymmetryCondition /. {ux → UX, ut → UT, utx → UTX, uxx → UXX, u[x, t] → U}
```

$$\begin{aligned}&UT\,\eta^{(0,0,1)}[x,t,U]-UXX\,\eta^{(0,0,1)}[x,t,U]+\eta^{(0,1,0)}[x,t,U]-\\&UX\left(UT\,\xi1^{(0,0,1)}[x,t,U]+\xi1^{(0,1,0)}[x,t,U]\right)-UT\left(UT\,\xi2^{(0,0,1)}[x,t,U]+\xi2^{(0,1,0)}[x,t,U]\right)+\\&2\,UXX\left(UX\,\xi1^{(0,0,1)}[x,t,U]+\xi1^{(1,0,0)}[x,t,U]\right)+\\&2\,UTX\left(UX\,\xi2^{(0,0,1)}[x,t,U]+\xi2^{(1,0,0)}[x,t,U]\right)-\\&UX\,\eta^{(1,0,1)}[x,t,U]-UX\left(UX\,\eta^{(0,0,2)}[x,t,U]+\eta^{(1,0,1)}[x,t,U]\right)-\\&\eta^{(2,0,0)}[x,t,U]+UX\left(UXX\,\xi1^{(0,0,1)}[x,t,U]+UX\,\xi1^{(1,0,1)}[x,t,U]+\right.\\&\quad\left.UX\left(UX\,\xi1^{(0,0,2)}[x,t,U]+\xi1^{(1,0,1)}[x,t,U]\right)+\xi1^{(2,0,0)}[x,t,U]\right)+\\&UT\left(UXX\,\xi2^{(0,0,1)}[x,t,U]+UX\,\xi2^{(1,0,1)}[x,t,U]+\right.\\&\quad\left.UX\left(UX\,\xi2^{(0,0,2)}[x,t,U]+\xi2^{(1,0,1)}[x,t,U]\right)+\xi2^{(2,0,0)}[x,t,U]\right)\end{aligned}$$

The equation is UT - UXX = 0. So UXX -> UT

```
SymmetryCondition = SymmetryCondition /. UXX → UT
```

$$\begin{aligned}&\eta^{(0,1,0)}[x,t,U]-UX\left(UT\,\xi1^{(0,0,1)}[x,t,U]+\xi1^{(0,1,0)}[x,t,U]\right)-\\&UT\left(UT\,\xi2^{(0,0,1)}[x,t,U]+\xi2^{(0,1,0)}[x,t,U]\right)+2\,UT\left(UX\,\xi1^{(0,0,1)}[x,t,U]+\xi1^{(1,0,0)}[x,t,U]\right)+\\&2\,UTX\left(UX\,\xi2^{(0,0,1)}[x,t,U]+\xi2^{(1,0,0)}[x,t,U]\right)-UX\,\eta^{(1,0,1)}[x,t,U]-\\&UX\left(UX\,\eta^{(0,0,2)}[x,t,U]+\eta^{(1,0,1)}[x,t,U]\right)-\eta^{(2,0,0)}[x,t,U]+\\&UX\left(UT\,\xi1^{(0,0,1)}[x,t,U]+UX\,\xi1^{(1,0,1)}[x,t,U]+UX\left(UX\,\xi1^{(0,0,2)}[x,t,U]+\xi1^{(1,0,1)}[x,t,U]\right)+\right.\\&\quad\left.\xi1^{(2,0,0)}[x,t,U]\right)+UT\left(UT\,\xi2^{(0,0,1)}[x,t,U]+UX\,\xi2^{(1,0,1)}[x,t,U]+\right.\\&\quad\left.UX\left(UX\,\xi2^{(0,0,2)}[x,t,U]+\xi2^{(1,0,1)}[x,t,U]\right)+\xi2^{(2,0,0)}[x,t,U]\right)\end{aligned}$$

```
SymmetryCondition = Collect[SymmetryCondition, {UX, UT, UTX}]
```

$$\begin{aligned}&UX^3\,\xi1^{(0,0,2)}[x,t,U]+\eta^{(0,1,0)}[x,t,U]+2\,UTX\,\xi2^{(1,0,0)}[x,t,U]+\\&UX^2\left(-\eta^{(0,0,2)}[x,t,U]+UT\,\xi2^{(0,0,2)}[x,t,U]+2\,\xi1^{(1,0,1)}[x,t,U]\right)-\\&\eta^{(2,0,0)}[x,t,U]+UX\left(2\,UTX\,\xi2^{(0,0,1)}[x,t,U]-\xi1^{(0,1,0)}[x,t,U]-2\,\eta^{(1,0,1)}[x,t,U]+\right.\\&\quad\left.UT\left(2\,\xi1^{(0,0,1)}[x,t,U]+2\,\xi2^{(1,0,1)}[x,t,U]\right)+\xi1^{(2,0,0)}[x,t,U]\right)+\\&UT\left(-\xi2^{(0,1,0)}[x,t,U]+2\,\xi1^{(1,0,0)}[x,t,U]+\xi2^{(2,0,0)}[x,t,U]\right)\end{aligned}$$

```
DeterminingEquations =
  DeleteCases[CoefficientList[SymmetryCondition, {UX, UT, UTX}] // Flatten, 0, {-1}]
```

$$\begin{aligned}&\left\{\eta^{(0,1,0)}[x,t,U]-\eta^{(2,0,0)}[x,t,U],\ 2\,\xi2^{(1,0,0)}[x,t,U],\right.\\&-\xi2^{(0,1,0)}[x,t,U]+2\,\xi1^{(1,0,0)}[x,t,U]+\xi2^{(2,0,0)}[x,t,U],\\&-\xi1^{(0,1,0)}[x,t,U]-2\,\eta^{(1,0,1)}[x,t,U]+\xi1^{(2,0,0)}[x,t,U],\\&2\,\xi2^{(0,0,1)}[x,t,U],\ 2\,\xi1^{(0,0,1)}[x,t,U]+2\,\xi2^{(1,0,1)}[x,t,U],\\&\left.-\eta^{(0,0,2)}[x,t,U]+2\,\xi1^{(1,0,1)}[x,t,U],\ \xi2^{(0,0,2)}[x,t,U],\ \xi1^{(0,0,2)}[x,t,U]\right\}\end{aligned}$$

DeterminingEquations // MatrixForm

$$\begin{pmatrix} \eta^{(0,1,0)}[x,t,U]-\eta^{(2,0,0)}[x,t,U] \\ 2\,\xi2^{(1,0,0)}[x,t,U] \\ -\xi2^{(0,1,0)}[x,t,U]+2\,\xi1^{(1,0,0)}[x,t,U]+\xi2^{(2,0,0)}[x,t,U] \\ -\xi1^{(0,1,0)}[x,t,U]-2\,\eta^{(1,0,1)}[x,t,U]+\xi1^{(2,0,0)}[x,t,U] \\ 2\,\xi2^{(0,0,1)}[x,t,U] \\ 2\,\xi1^{(0,0,1)}[x,t,U]+2\,\xi2^{(1,0,1)}[x,t,U] \\ -\eta^{(0,0,2)}[x,t,U]+2\,\xi1^{(1,0,1)}[x,t,U] \\ \xi2^{(0,0,2)}[x,t,U] \\ \xi1^{(0,0,2)}[x,t,U] \end{pmatrix}$$

```
ξ1[x_, t_, U_] = A1[x, t] U + A2[x, t];
ξ2[x_, t_, U_] = B[t];
```

```
DeterminingEquations2 =
  DeleteCases[CoefficientList[SymmetryCondition, {UX, UT, UTX}] // Flatten, 0, {-1}]
```

$$\left\{\eta^{(0,1,0)}[x,t,U]-\eta^{(2,0,0)}[x,t,U],\ -B'[t]+2\left(U\,A1^{(1,0)}[x,t]+A2^{(1,0)}[x,t]\right),\ -U\,A1^{(0,1)}[x,t]-A2^{(0,1)}[x,t]+U\,A1^{(2,0)}[x,t]+A2^{(2,0)}[x,t]-2\,\eta^{(1,0,1)}[x,t,U],\ 2\,A1[x,t],\ 2\,A1^{(1,0)}[x,t]-\eta^{(0,0,2)}[x,t,U]\right\}$$

DeterminingEquations2 // MatrixForm

$$\begin{pmatrix} \eta^{(0,1,0)}[x,t,U]-\eta^{(2,0,0)}[x,t,U] \\ -B'[t]+2\left(U\,A1^{(1,0)}[x,t]+A2^{(1,0)}[x,t]\right) \\ -U\,A1^{(0,1)}[x,t]-A2^{(0,1)}[x,t]+U\,A1^{(2,0)}[x,t]+A2^{(2,0)}[x,t]-2\,\eta^{(1,0,1)}[x,t,U] \\ 2\,A1[x,t] \\ 2\,A1^{(1,0)}[x,t]-\eta^{(0,0,2)}[x,t,U] \end{pmatrix}$$

```
A1[x_, t_] = 0;
```

```
DeterminingEquations3 =
  DeleteCases[CoefficientList[SymmetryCondition, {UX, UT, UTX}] // Flatten, 0, {-1}]
```

$$\left\{\eta^{(0,1,0)}[x,t,U]-\eta^{(2,0,0)}[x,t,U],\ -B'[t]+2\,A2^{(1,0)}[x,t],\ -A2^{(0,1)}[x,t]+A2^{(2,0)}[x,t]-2\,\eta^{(1,0,1)}[x,t,U],\ -\eta^{(0,0,2)}[x,t,U]\right\}$$

DeterminingEquations3 // MatrixForm

$$\begin{pmatrix} \eta^{(0,1,0)}[x,t,U]-\eta^{(2,0,0)}[x,t,U] \\ -B'[t]+2\,A2^{(1,0)}[x,t] \\ -A2^{(0,1)}[x,t]+A2^{(2,0)}[x,t]-2\,\eta^{(1,0,1)}[x,t,U] \\ -\eta^{(0,0,2)}[x,t,U] \end{pmatrix}$$

```
η[x_, t_, U_] = C1[x, t] U + C2[x, t];
```

DSolve$\left[-B'[t]+2\,A2^{(1,0)}[x,t]==0,\ A2,\ \{x,t\}\right]$

$$\left\{\left\{A2\to \text{Function}\left[\{x,t\},\ C[1][t]+\frac{1}{2}\,x\,B'[t]\right]\right\}\right\}$$

```
A2[x_, t_] = A21[t] + 1/2 x B′[t];
```

```
DeterminingEquations4 =
 DeleteCases[CoefficientList[SymmetryCondition, {UX, UT, UTX}] // Flatten, 0, {-1}]
```

$$\left\{U\,C1^{(0,1)}[x,t] + C2^{(0,1)}[x,t] - U\,C1^{(2,0)}[x,t] - C2^{(2,0)}[x,t],\; -A21'[t] - \frac{1}{2}x\,B''[t] - 2\,C1^{(1,0)}[x,t]\right\}$$

```
DeterminingEquations4 // MatrixForm
```

$$\begin{pmatrix} U\,C1^{(0,1)}[x,t] + C2^{(0,1)}[x,t] - U\,C1^{(2,0)}[x,t] - C2^{(2,0)}[x,t] \\ -A21'[t] - \frac{1}{2}x\,B''[t] - 2\,C1^{(1,0)}[x,t] \end{pmatrix}$$

```
DSolve[-A21′[t] - 1/2 x B″[t] - 2 C1^(1,0)[x, t] == 0, C1, {x, t}]
```

$$\left\{\left\{C1 \to \text{Function}\left[\{x,t\},\, C[1][t] + \frac{1}{4}\left(-2\,x\,A21'[t] - \frac{1}{2}x^2\,B''[t]\right)\right]\right\}\right\}$$

```
C1[x_, t_] = C11[t] + 1/4 (-2 x A21′[t] - 1/2 x^2 B″[t]);
SymmetryCondition = Collect[SymmetryCondition // FullSimplify, {U}]
```

$$\frac{1}{8}U\left(8\,C11'[t] + 2\,B''[t] - x\left(4\,A21''[t] + x\,B^{(3)}[t]\right)\right) + C2^{(0,1)}[x,t] - C2^{(2,0)}[x,t]$$

```
eq1 = Collect[8 Coefficient[SymmetryCondition, U], x]
```

$$8\,C11'[t] - 4\,x\,A21''[t] + 2\,B''[t] - x^2\,B^{(3)}[t]$$

```
B[t_] = c1 + c2 t + c3 t^2;
A21[t_] = c4 + c5 t;
```

```
eq1
```

$$4\,c3 + 8\,C11'[t]$$

```
DSolve[eq1 == 0, C11, t]
```

$$\left\{\left\{C11 \to \text{Function}\left[\{t\},\, -\frac{c3\,t}{2} + C[1]\right]\right\}\right\}$$

```
C11[t_] = -c3 t/2 + c6
```

$$c6 - \frac{c3\,t}{2}$$

```
Symmetries = {ξ1[x, t, u], ξ2[x, t, u], η[x, t, u]}
SymmetryCondition = SymmetryCondition // FullSimplify
```

$$\left\{c4 + c5\,t + \frac{1}{2}(c2 + 2\,c3\,t)\,x,\; c1 + c2\,t + c3\,t^2,\; u\left(c6 - \frac{c3\,t}{2} + \frac{1}{4}\left(-2\,c5\,x - c3\,x^2\right)\right) + C2[x,t]\right\}$$

$$C2^{(0,1)}[x,t] - C2^{(2,0)}[x,t]$$

```
Symmetries /. {c1 → 1, c2 → 0, c3 → 0, c4 → 0, c5 → 0, c6 → 0, C2[x, t] → 0}
Symmetries /. {c1 → 0, c2 → 2, c3 → 0, c4 → 0, c5 → 0, c6 → 0, C2[x, t] → 0}
Symmetries /. {c1 → 0, c2 → 0, c3 → 1, c4 → 0, c5 → 0, c6 → 0, C2[x, t] → 0}
Symmetries /. {c1 → 0, c2 → 0, c3 → 0, c4 → 1, c5 → 0, c6 → 0, C2[x, t] → 0}
Symmetries /. {c1 → 0, c2 → 0, c3 → 0, c4 → 0, c5 → 1, c6 → 0, C2[x, t] → 0}
Symmetries /. {c1 → 0, c2 → 0, c3 → 0, c4 → 0, c5 → 0, c6 → 1, C2[x, t] → 0}

Symmetries /. {c1 → 0, c2 → 0, c3 → 0, c4 → 0, c5 → 0, c6 → 0}
```

$\{0, 1, 0\}$

$\{x, 2t, 0\}$

$\left\{t x, t^2, u\left(-\frac{t}{2} - \frac{x^2}{4}\right)\right\}$

$\{1, 0, 0\}$

$\left\{t, 0, -\frac{u x}{2}\right\}$

$\{0, 0, u\}$

$\{0, 0, C2[x, t]\}$